**************************************************************************************
************************* Table A.2: Summary Statistics ******************************
**************************************************************************************
use data_final_for_appendix, clear

global sum total_cases_per_million total_deaths_per_million stdtr v2x_liberal v2x_polyarchy tax wbgi_gee wbgi_rqe leaderage female length_ce college major_n_2 major_n_3 major_n_4 major_n_5 TBI /// 
dis_wuhan dis_milan rainfally temperaturey island ter_ruggedness gdpp pop life_expectancy aged_65_older urban YSchool ghs dalys Route if cross_samp==1 & ///
!missing(total_cases_per_million, total_deaths_per_million, stdtr, v2x_liberal, v2x_polyarchy, tax, wbgi_gee, wbgi_rqe, leaderage, female, length_ce, college, TBI, major_n_2, major_n_3, major_n_4, major_n_5, dis_wuhan, dis_milan, rainfally, temperaturey, island, ter_ruggedness, gdpp, pop, life_expectancy, aged_65_older, urban, YSchool, ghs, dalys, Route)

eststo summ: estpost summarize $sum 

esttab summ using Table_A2.txt, booktabs ///
refcat(total_cases_per_million "\emph{Pandemic Performance}" stdtr "\emph{Political Trust}" v2x_liberal "Regime Type" tax "\emph{State Capacity}" leaderage "\emph{Leader Attributes}" dis_wuhan "\emph{Other Covariates}", nolabel) ///
nomtitle nonumber noobs cells("mean(fmt(a3)) sd(fmt(a3)) min(fmt(a3)) max(fmt(a3))") replace label

**************************************************************************************
******************************Appendix C: Robutness Check*****************************
**************************************************************************************
global ctrls c_*
global regime r_*
global capacity cp_*
global leader ld_*

**# Table A.4: Robustness: Controlling for Cultural Cohesion
use data_final_for_appendix, clear

eststo clear
// Confirmed cases per million //
reghdfe log_case_cumtotal_per1m stdtr std_interperson_full $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m1 
 
reghdfe log_case_cumtotal_per1m stdtr std_socialtrust $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m2

reghdfe log_case_cumtotal_per1m stdtr std_socialcapital $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m3

// Confirmed deaths per million //
reghdfe log_death_cumtotal_per1m stdtr std_interperson_full $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m4

reghdfe log_death_cumtotal_per1m stdtr std_socialtrust $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m5

reghdfe log_death_cumtotal_per1m stdtr std_socialcapital $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m6

label variable std_interperson_full "Interpersonal Trust"
label variable std_socialtrust "Social Trust (WVS)"
label variable std_socialcapital "Social Capital (WVS)"

esttab m1 m2 m3 m4 m5 m6 using Table_A4.txt, replace ///
title("Robustness: Controlling for Cultural Cohesion" \label{check:culturalcohesion}) ///
mgroups("Cases" "Deaths", pattern(1 0 0 1) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span})) ///
booktabs b(3) se(3) nonote nocons nobaselevels nodepvars ar2 ///
order(stdtr std_interperson_full std_socialtrust std_socialcapital) ///
drop($regime $capacity ld_leaderage ld_female ld_length_ce ld_college ld_major_n_3 ld_major_n_4 ld_major_n_5 ld_TBI $ctrls _cons) nogaps ///
label s(EDCov fixedm N r2_a, label("Covariates" "Regional FE" "Observations" "Adjusted R$^2$") fmt(0 0 0 3)) ///
compress nomtitles starlevels(* .1 ** 0.05 *** 0.01)

**# Table A.5: Robustness: Controlling for Corruption
use data_final_for_appendix, clear

eststo clear
// Confirmed cases per million //
reghdfe log_case_cumtotal_per1m stdtr bci_bci $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m1

reghdfe log_case_cumtotal_per1m stdtr ti_cpi $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m2

reghdfe log_case_cumtotal_per1m stdtr vdem_corr $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m3

reghdfe log_case_cumtotal_per1m stdtr wbgi_cce $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m4

// Confirmed deaths per million //
reghdfe log_death_cumtotal_per1m stdtr bci_bci $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m5

reghdfe log_death_cumtotal_per1m stdtr ti_cpi $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m6

reghdfe log_death_cumtotal_per1m stdtr vdem_corr $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m7

reghdfe log_death_cumtotal_per1m stdtr wbgi_cce $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m8

label variable bci_bci "Bayesian Corruption Index"
label variable ti_cpi "Corruption Perception"
label variable vdem_corr "Political Corruption"
label variable wbgi_cce "Control of Corruption"

esttab m1 m2 m3 m4 m5 m6 m7 m8 using Table_A5.txt, replace ///
title("Robustness: Controlling for Corruption" \label{check:corruption}) ///
mgroups("Cases" "Deaths", pattern(1 0 0 0 1) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span})) ///
booktabs b(3) se(3) nonote nocons nobaselevels nodepvars ar2 ///
order(stdtr bci_bci ti_cpi vdem_corr wbgi_cce) ///
drop($regime $capacity ld_leaderage ld_female ld_length_ce ld_college ld_major_n_3 ld_major_n_4 ld_major_n_5 ld_TBI $ctrls _cons) nogaps ///
label s(EDCov fixedm N r2_a, label("Covariates" "Regional FE" "Observations" "Adjusted R$^2$") fmt(0 0 0 3)) ///
compress nomtitles starlevels(* .1 ** 0.05 *** 0.01)

**# Table A.6: Robustness: Controlling for Social Cleavages
use data_final_for_appendix, clear

eststo clear

reghdfe log_case_cumtotal_per1m stdtr $regime $capacity $leader $ctrls ctrl_* if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m1

dsregress log_case_cumtotal_per1m stdtr if cross_samp==1, controls((i.ht_region) $regime $capacity $leader $ctrls ctrl_*) selection(cv) vce(robust) rseed(12345)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m2

// LASSO selection results //
estimates store ds_cv
lassocoef(ds_cv, for (stdtr))

reghdfe log_death_cumtotal_per1m stdtr $regime $capacity $leader $ctrls ctrl_* if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m3

dsregress log_death_cumtotal_per1m stdtr if cross_samp==1, controls((i.ht_region) $regime $capacity $leader $ctrls ctrl_*) selection(cv) vce(robust) rseed(12345)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m4

// LASSO selection results //
estimates store ds_cv
lassocoef(ds_cv, for (stdtr))

label variable ctrl_efindex "Ethnic Fractionalization"
label variable ctrl_dom_relig_pct "Dominant Religion Share"
label variable ctrl_top_top1_income_share "Top 1\% Income Share"
label variable ctrl_top_top10_income_share "Top 10\% Income Share"

esttab m1 m2 m3 m4 using Table_A6.txt, replace ///
title("Robustness: Controlling for Social Cleavages" \label{check:cleavage}) ///
mgroups("Cases" "Deaths", pattern(1 0 1 0) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span})) ///
booktabs b(3) se(3) nonote nocons nobaselevels nodepvars ar2 ///
order(stdtr ctrl_efindex ctrl_dom_relig_pct ctrl_top_top1_income_share ctrl_top_top10_income_share) ///
drop($regime $capacity ld_leaderage ld_female ld_length_ce ld_college ld_major_n_3 ld_major_n_4 ld_major_n_5 ld_TBI $ctrls _cons) nogaps ///
label s(EDCov fixedm N r2_a, label("Covariates" "Regional FE" "Observations" "Adjusted R$^2$") fmt(0 0 0 3)) ///
compress nomtitles starlevels(* .1 ** 0.05 *** 0.01)

**# Table A.7: Robustness: Effects of Trust in Executive Branch
use data_final_for_appendix, clear

eststo clear
// Confirmed cases per million //
reghdfe log_case_cumtotal_per1m stdtr_gov if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
est store m1

reghdfe log_case_cumtotal_per1m stdtr_gov $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m2

reghdfe log_case_cumtotal_per1m stdtr_gov $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m3

dsregress log_case_cumtotal_per1m stdtr_gov if cross_samp==1, controls((i.ht_region) $regime $capacity $leader $ctrls) selection(cv) vce(robust) rseed(12345)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m4

// LASSO selection results //
estimates store ds_cv
lassocoef(ds_cv, for (stdtr_gov))

// Confirmed deaths per million //
reghdfe log_death_cumtotal_per1m stdtr_gov if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
est store m5

reghdfe log_death_cumtotal_per1m stdtr_gov $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m6

reghdfe log_death_cumtotal_per1m stdtr_gov $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m7

dsregress log_death_cumtotal_per1m stdtr_gov if cross_samp==1, controls((i.ht_region) $regime $capacity $leader $ctrls) selection(cv) vce(robust) rseed(12345)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m8

// LASSO selection results //
estimates store ds_cv
lassocoef(ds_cv, for (stdtr_gov))

label variable stdtr_gov "Trust in Executive Branch"

esttab m1 m2 m3 m4 m5 m6 m7 m8 using Table_A7.txt, replace ///
title("Robustness: Effects of Trust in Executive Branch" \label{check:executivetrust}) ///
mgroups("Cases" "Deaths", pattern(1 0 0 0 1) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span})) ///
booktabs b(3) se(3) nonote nocons nobaselevels nodepvars ar2 ///
order(stdtr_gov $regime $capacity ld_leaderage ld_female ld_length_ce ld_college ld_major_n_3 ld_major_n_4 ld_major_n_5 ld_TBI) ///
drop($ctrls _cons) nogaps ///
label s(EDCov fixedm N r2_a, label("Geog \& Socioecon. Covariates" "Regional FE" "Observations" "Adjusted R$^2$") fmt(0 0 0 3)) ///
compress nomtitles starlevels(* .1 ** 0.05 *** 0.01)

**# Table A.8: Robustness: Effects of Political Trust on Excess Deaths
use data_final_for_appendix, clear

eststo clear

gen excess_mean_per1m = excess_mean_e5*10
gen log_excess_mean_per1m = log(excess_mean_per1m+2461) //min is -2460

reghdfe log_excess_mean_per1m stdtr if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
est store m1

reghdfe log_excess_mean_per1m stdtr $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m2

reghdfe log_excess_mean_per1m stdtr $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m3

dsregress log_excess_mean_per1m stdtr if cross_samp==1, controls((i.ht_region) $regime $capacity $leader $ctrls) selection(cv) vce(robust) rseed(12345)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m4

// LASSO selection results //
estimates store ds_cv
lassocoef(ds_cv, for (stdtr))

esttab m1 m2 m3 m4 using Table_A8.txt, replace ///
title("Robustness: Effects of Political Trust on Excess Deaths" \label{check:Excess_death}) ///
booktabs b(3) se(3) nonote nocons nobaselevels nodepvars ar2 ///
order(stdtr $regime $capacity ld_leaderage ld_female ld_length_ce ld_college ld_major_n_3 ld_major_n_4 ld_major_n_5 ld_TBI) ///
drop($ctrls _cons) nogaps ///
label s(EDCov fixedm N r2_a, label("Geog \& Socioecon. Covariates" "Regional FE" "Observations" "Adjusted R$^2$") fmt(0 0 0 3)) ///
compress nomtitles starlevels(* .1 ** 0.05 *** 0.01) 

**# Figure A.8: Robustness: Effects of Political Trust for Different Dates (Mar 2020 - Jan 2021)
use data_final_for_appendix, clear

clear matrix
mat D=J(346,5,.)
levelsof date if total_cases_per_million!=., local(lv)
forv i=1/346{
local j : word `i' of `lv' 

capture drop dv
g dv=log(total_cases_per_million+1) if date==`j'

capture reghdfe dv stdtr $regime $capacity $leader $ctrls if date==`j' ,a(ht_region) vce(robust) 

capture mat D[`i',1]=_b[stdtr]
capture mat D[`i',2]=_b[stdtr]-1.96*_se[stdtr]
capture mat D[`i',3]=_b[stdtr]+1.96*_se[stdtr]
capture mat D[`i',4]=e(N)
capture mat D[`i',5]=`j'
} 

matrix list D

mat2txt, matrix(D) saving(diffdate_case) replace

clear matrix
mat D=J(346,5,.)
levelsof date if total_deaths_per_million!=., local(lv)
forv i=1/346{
local j : word `i' of `lv' 

capture drop dv
g dv=log(total_deaths_per_million+1) if date==`j'

capture reghdfe dv stdtr $regime $capacity $leader $ctrls if date==`j' ,a(ht_region) vce(robust) 

capture mat D[`i',1]=_b[stdtr]
capture mat D[`i',2]=_b[stdtr]-1.96*_se[stdtr]
capture mat D[`i',3]=_b[stdtr]+1.96*_se[stdtr]
capture mat D[`i',4]=e(N)
capture mat D[`i',5]=`j'
} 

matrix list D

mat2txt, matrix(D) saving(diffdate_death) replace

**# Figure A.9: Robustness: Effects of Political Trust from Previous Years (2009-2018)
clear 
clear matrix
mat D=J(10,3,.)

local m 1
forvalues y=2009/2018{
use data_constructed_trust, clear
keep if year == `y'

merge 1:m ccodealp using data_final_for_appendix, nogenerate
capture reghdfe log_case_cumtotal_per1m stdtr $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)

capture mat D[`m',1]=_b[stdtr]
capture mat D[`m',2]=_b[stdtr]-1.96*_se[stdtr]
capture mat D[`m',3]=_b[stdtr]+1.96*_se[stdtr]

local m=`m'+1
}

matrix list D
mat2txt, matrix(D) saving(diffyear_case) replace

clear 
clear matrix
mat D=J(10,3,.)

local m 1
forvalues y=2009/2018{
use data_constructed_trust, clear
keep if year == `y'

merge 1:m ccodealp using data_final_for_appendix, nogenerate
capture reghdfe log_death_cumtotal_per1m stdtr $regime $capacity $leader $ctrls if cross_samp==1, a(ht_region) vce(robust)

capture mat D[`m',1]=_b[stdtr]
capture mat D[`m',2]=_b[stdtr]-1.96*_se[stdtr]
capture mat D[`m',3]=_b[stdtr]+1.96*_se[stdtr]

local m=`m'+1
}

matrix list D
mat2txt, matrix(D) saving(diffyear_death) replace


**# Table A.9: Robustness: Effects of Political Trust Excluding Certain Regions (Cases)
use data_final_for_appendix, clear

eststo clear
*Remove one region
*Excluding Eastern Europe and Post-Soviet countries
reghdfe log_case_cumtotal_per1m c.stdtr $regime $capacity $leader $ctrls if cross_samp==1 & ht_region!=1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local Cov "Yes", replace
est store m1

*Latin America
reghdfe log_case_cumtotal_per1m c.stdtr $regime $capacity $leader $ctrls if cross_samp==1 & ht_region!=2, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local Cov "Yes", replace
est store m2

*North Africa and Middle East
reghdfe log_case_cumtotal_per1m c.stdtr $regime $capacity $leader $ctrls if cross_samp==1 & ht_region!=3, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local Cov "Yes", replace
est store m3

*Sub-Saharan Africa
reghdfe log_case_cumtotal_per1m c.stdtr $regime $capacity $leader $ctrls if cross_samp==1 & ht_region!=4, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local Cov "Yes", replace
est store m4

*Western Europe and North America
reghdfe log_case_cumtotal_per1m c.stdtr $regime $capacity $leader $ctrls if cross_samp==1 & ht_region!=5, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local Cov "Yes", replace
est store m5

*East Asia
reghdfe log_case_cumtotal_per1m c.stdtr $regime $capacity $leader $ctrls if cross_samp==1 & ht_region!=6, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local Cov "Yes", replace
est store m6

*South-East Asia
reghdfe log_case_cumtotal_per1m c.stdtr $regime $capacity $leader $ctrls if cross_samp==1 & ht_region!=7, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local Cov "Yes", replace
est store m7

*South Asia
reghdfe log_case_cumtotal_per1m c.stdtr $regime $capacity $leader $ctrls if cross_samp==1 & ht_region!=8, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local Cov "Yes", replace
est store m8

*the Pacific
reghdfe log_case_cumtotal_per1m c.stdtr $regime $capacity $leader $ctrls if cross_samp==1 & ht_region!=9, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local Cov "Yes", replace
est store m9

*the Caribbean
reghdfe log_case_cumtotal_per1m c.stdtr $regime $capacity $leader $ctrls if cross_samp==1 & ht_region!=10, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local Cov "Yes", replace
est store m10

esttab m1 m2 m3 m4 m5 m6 m7 m8 m9 m10 using Table_A9.txt, replace ///
title("Robustness: Effects of Political Trust Excluding Certain Regions (Cases)" \label{check:Region}) ///
booktabs b(3) se(3) nonote nocons nobaselevels nodepvars ar2 ///
keep(stdtr) nomtitles ///
label s(Cov fixedm N r2_a, label("Covariates" "Regional FE" "Observations" "Adjusted R$^2$") fmt(0 0 0 3 0)) ///
compress nomtitles starlevels(* .1 ** 0.05 *** 0.01)

**# Table A.10: Robustness: Effect of Political Trust (Excluding the Highest and Lowest 10% of the Sample)
use data_final_for_appendix, clear

eststo clear

reghdfe log_case_cumtotal_per1m c.stdtr $regime $capacity $leader $ctrls if cross_samp==1 & (log_case_cumtotal_per1m<10.80989 & log_case_cumtotal_per1m>5.86574), a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local Cov "Yes", replace
est store m1

reghdfe log_death_cumtotal_per1m c.stdtr $regime $capacity $leader $ctrls if cross_samp==1 & (log_death_cumtotal_per1m<6.889781 & log_death_cumtotal_per1m>1.847352), a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local Cov "Yes", replace
est store m2

esttab m1 m2 using Table_A10.txt, replace ///
title("Robustness: Effect of Political Trust (Excluding the Highest and Lowest 10\% of the Sample)" \label{check:Outlier}) ///
booktabs b(3) se(3) nonote nocons nobaselevels nodepvars ar2 ///
mtitles("Cases" "Deaths") ///
keep(stdtr) ///
label s(Cov fixedm N r2_a, label("Covariates" "Regional FE" "Observations" "Adjusted R$^2$") fmt(0 0 0 3)) ///
compress starlevels(* .1 ** 0.05 *** 0.01)

**# Table A.11: Robustness: Controlling for Missing Values 
use data_final_for_appendix, clear

eststo clear
replace stdtr=99 if stdtr==.
gen missing_trust=1 if stdtr==99
replace missing_trust=0 if stdtr!=99

reghdfe log_case_cumtotal_per1m stdtr $regime $capacity $leader $ctrls missing_trust if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m1

dsregress log_case_cumtotal_per1m stdtr if cross_samp==1, controls((i.ht_region missing_trust) $regime $capacity $leader $ctrls) selection(cv) vce(robust) rseed(12345)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m2

// LASSO selection results //
estimates store ds_cv
lassocoef(ds_cv, for (stdtr))

reghdfe log_death_cumtotal_per1m stdtr $regime $capacity $leader $ctrls missing_trust if cross_samp==1, a(ht_region) vce(robust)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m3

dsregress log_death_cumtotal_per1m stdtr if cross_samp==1, controls((i.ht_region missing_trust) $regime $capacity $leader $ctrls) selection(cv) vce(robust) rseed(12345)
quietly estadd local fixedm "Yes", replace
quietly estadd local EDCov "Yes", replace
est store m4

// LASSO selection results //
estimates store ds_cv
lassocoef(ds_cv, for (stdtr))

label variable missing_trust "Missing Trust"

esttab m1 m2 m3 m4 using Table_A11.txt, replace ///
title("Robustness: Controlling for Missing Values" \label{check:Missingtrust}) ///
mgroups("Cases" "Deaths", pattern(1 0 1 0) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span})) ///
booktabs b(3) se(3) nonote nocons nobaselevels nodepvars ar2 ///
order(stdtr $regime $capacity ld_leaderage ld_female ld_length_ce ld_college ld_major_n_3 ld_major_n_4 ld_major_n_5 ld_TBI) ///
drop($ctrls _cons) nogaps ///
label s(EDCov fixedm N r2_a, label("Geog \& Socioecon. Covariates" "Regional FE" "Observations" "Adjusted R$^2$") fmt(0 0 0 3)) ///
compress nomtitles starlevels(* .1 ** 0.05 *** 0.01)